

## A Quick Patient Profile: Combining External Data with EDC-generated Subject CRF

Titania Dumas-Roberson, Grifols Therapeutics, Inc., Durham, NC  
Yang Han, Grifols Therapeutics, Inc., Durham, NC

### ABSTRACT

This paper reads the current PDF files within a directory to generate a list of subjects which need their clinical data combined with external vendor data. It creates dynamic Visual Basic script which is used to merge multiple PDFs. The VB script is then executed to produce a combined PDF with active bookmarks.

SAS® 9.3 and Adobe® Acrobat® 9 Pro were utilized to develop this program.

### INTRODUCTION

With the increased use of electronic data capture (EDC) systems for clinical trials, it has become easier to provide a subject's entire case report form (CRF) with just a few clicks. However, this EDC-generated CRF only provides data present in the clinical database. Oftentimes, when a Medical Monitor needs to write a serious adverse event (SAE) narrative, the data collected within the EDC system does not suffice. Additional supporting information is sometimes stored in external data files, such as central lab or imaging data sets.

Some benefits to using the EDC-generated include:

1. Limited programming required
2. Data presented is up-to-date
3. Access to the fields captured, audit trail and comments entered for items for end-user

### PARSE THE DIRECTORY

We begin by defining a few user-supplied parameters, to make the code transferable between various projects.

```
%LET profiledir=C:\PharmaSUG PDF\ ;  
%LET external=C:\PharmaSUG PDF\Central Lab\ ;  
LIBNAME clabdata "&external" ACCESS=readonly ;
```

Since the study may have hundreds of subjects, the first step is to select only the subject(s) who need an updated patient profile. The subject CRFs from the EDC system should be stored in a specified directory. A DATA step is used to search the directory.

```
FILENAME direct pipe "dir "&profiledir.*.*" " ;  
  
DATA dirlist (keep=subj_num) ;  
  INFILE direct lrecl=200 trunccover ;  
  INPUT line $200. ; LENGTH subj_num $15 ;  
  subj_num=upcase(scan(line,-1," ")) ;  
  if index(subj_num, '.PDF')=0 then delete ;  
  subj_num=scan(subj_num,1, '.') ;  
RUN ;
```

For processing ease and flexibility, each subject within the directory and the total number of subjects are assigned to a macro variable. Additionally, the columns needed to display from the external data set were selected and sorted to allow for desired pagination.

### GENERATE A TEMPORARY PDF FOR EXTERNAL DATA

Some external vendors provide reports in PDF format, but access to these reports by a statistical programmer can be limited or possibly outdated. In order to circumvent delays by other parties, one solution is to generate a temporary PDF which contains the most recently transferred data from the external vendor. This is achieved by combination of the PRINT procedure and an ODS PDF statement.

Bookmarking is important to the end-user for navigation, therefore options PDFTOC, BOOKMARKGEN and BOOKMARKLIST are incorporated.

```
ODS PDF file="&profiledir.\&&Subj&subno. Lab.pdf" PDFTOC=1 BOOKMARKGEN=yes
BOOKMARKLIST=show startpage=yes ;

ODS PROCLABEL="CLAB (Central Lab Results)" ;
PROC PRINT data=CLAB HEADING=horizontal LABEL WIDTH=min ;
  WHERE subjid="&&subj&subno" ;
  BY battnam lbtest ;
  PAGEBY lbtest ;
RUN ;
```

Repeat the generation of PDFs, as needed, for each subject and external data source.

## CREATE VISUAL BASIC CODE TO APPEND PDFS

In order to append the PDFs, SAS needs to interact with Adobe using OLE automation. Programmatically, this is accomplished by dynamically generating a data set containing VB script and order of execution. The VB code opens the PDFs which need to be combined, then inserts then pages from the secondary PDF after the final page of the primary PDF. The method *InsertPages()* is important code as it contains information crucial to proper file insertion. Its syntax follows:

*InsertPages(nInsertPageAfter, iPDDocSource, nStartPage, nNumPages, bBookmarks)* where *bBookmarks* must be a positive number for bookmarks to be copied into the combined file.

VBScripts	VBOrder
Dim eCRFPdfDoc	1
Dim CLABPdfDoc	1
Set eCRFPdfDoc= CreateObject("AcroExch.PDDoc")	2
Set CLABPdfDoc= CreateObject("AcroExch.PDDoc")	2
file1=eCRFPdfDoc.Open("C:\PharmaSUG PDF\1991234.pdf")	3
file2=CLABPdfDoc.Open("C:\PharmaSUG PDF\1991234 Lab.pdf")	3
Stack = eCRFPdfDoc.InsertPages(eCRFPdfDoc.GetNumPages-1,CLABPdfDoc,0,CLABPdfdoc.GetNumPages,1)	4
SaveStack = eCRFPdfDoc.Save(1,"C:\PharmaSUG PDF\1991234 Combo.pdf")	5

**Table 1. Displays the contents of the dataset containing Visual Basic commands created for the sample subject.**

Refer to the listed reference for additional detailed information on the VB functions and their meanings.

## SUBMIT THE VBS FILE


Once the data set has been generated, the next steps are to produce and submit a VBScript file.

```
FILENAME vbsout "&profiledir.PDF_Combine.vbs" ;

DATA generate ;
  SET getstmt ;
  FILE vbsout ;
  PUT vbscripts ;
RUN ;

OPTIONS XSYNC NOXWAIT ;
X %unquote( %str('%') "&profiledir.PDF_Combine.vbs" %str('%') ) ;
```

The system option XSYNC is utilized to insure that X command completes execution prior to returning control back to the SAS session, while NOXWAIT dictates the command prompt window initiated by the X command automatically returns to the SAS session.



### Display 1. Final combined PDF output

To finish the process, the temporary lab files are deleted from the directory.

```
X %unquote( %str('% del) "&profiledir.&&subj&subno Lab.pdf" %str('%') ) ;
```

## CONCLUSION

Since EDC systems have the capability of generating the entire subject CRF, the need to independently program a patient profile is limited. Incorporation of external data can be achieved by appending the data to the EDC-generated PDF with a few common blocks of programming code.

## REFERENCES

Welch, Brandon. 2011. "Combining External PDF Files by Integrating SAS® and Adobe® Acrobat" *Proceedings of the Southeast SAS User's Group Conference 2011*. Raleigh, NC: Institute for Advanced Analytics, NCSU. Available at <http://analytics.ncsu.edu/sesug/2011/BB15.Welch.pdf>.

## ACKNOWLEDGMENTS

We would like to thank Peg Becker for providing dummy data for this paper, as well as our families for their continued support.

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Titania Dumas-Roberson  
 Enterprise: Grifols Therapeutics, Inc.  
 Address: 79 TW Alexander, 4101 Research Commons Drive  
 City, State ZIP: Durham, NC 27709  
 Work Phone: (919) 316-6153  
 E-mail: Titania.Dumas-Roberson@grifols.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.

## APPENDIX

```

*****
BEGIN: User-supplied parameters
***** ;
%LET profiledir=C:\PharmaSUG PDF\ ; **List storage location for EDC-generated CRFs** ;
%LET EXTERNAL=C:\PharmaSUG PDF\Central Lab\ ; ** List location of data ** ;
%LET lab_dt=2014-01-27 ; ** Date of external lab transfer ** ;
*****
END: User-supplied parameters
***** ;

LIBNAME clabdata "&external" ACCESS=readonly ;
FILENAME direct pipe "dir "&profiledir.*.*" " ;

*****
Read all files in directory, keep list with extension .PDF
***** ;
DATA dirlist (keep=subj_num) ;
  infile direct lrecl=200 truncover ;
  input line $200. ;
  length subj_num $15 ;
  subj_num=upcase(scan(line,-1," ")) ;
  if index(subj_num, '.PDF')=0 then delete ;
  subj_num=scan(subj_num,1, '.') ;
RUN ;

*****
Assign macro variables to each subject in directory
***** ;
PROC SQL NOPRINT ;
  SELECT put( count(distinct subj_num),2.) into :totsubj from dirlist ;
  SELECT distinct subj_num into :subj1 - :subj%cmpres(&totsubj) from dirlist order
  subj_num ;
  CREATE table clab as select subjid, visit, visittyp, battnam, lbtest, rptresc,
  rptu, rptnrlo, rptnrhi, alrtfl
  from clabdata.clab order subjid, battnam, lbtest, lbdtm ;
QUIT ;

OPTION ORIENTATION=portrait TOPMARGIN=1in LEFTMARGIN=1in RIGHTMARGIN=1in NOCENTER
NODATE NONUMBER ;
ODS LISTING CLOSE ;

%macro loopsubjs ;

  %do subno = 1 %to &totsubj ;
  *****
  Build temporary CRF with central lab data
  ***** ;
  ODS PDF FILE="&profiledir.&&Subj&subno. Lab.pdf" PDFTOC=1 BOOKMARKGEN=yes
  BOOKMARKLIST=show STARTPAGE=yes ;

  title1 j=1 "Study ID: TST2013-05" color=blue j=c bold bcolor=yellow "Subject
  &&subj&subno" ;
  footnote1 j=1 "Data extracted on: &lab_dt" ;

  ***** print the actual data ***** ;
  ODS PROCLABEL="CLAB (Central Lab Results)" ;
  PROC PRINT DATA=CLAB HEADING=horizontal LABEL WIDTH=min ;
  title2 j=1 "Dataset: CLAB (Central Lab Results)" ;
  WHERE subjid="&&subj&subno" ;
  BY battnam lbtest ;
  PAGEBY lbtest ;

```

```

RUN ;

ODS PDF CLOSE ;

*****
Write Visual Basic code to append PDFs
***** ;
DATA getstmt ;
  DO type='eCRFPdfDoc','CLABPdfDoc' ;
    LENGTH VBScripts $200 ;
    vbscripts='Dim '||strip(type) ;
    VBOOrder=1 ;
    OUTPUT ;
    vbscripts='Set '||strip(type)||'= CreateObject("AcroExch.PDDoc")' ;
    VBOOrder=2 ;
    OUTPUT ;
    if type='eCRFPdfDoc' then
vbscripts='file1='||strip(type)||'.Open("||"&profiledir.&&Subj&subno"||'.pdf)" ' ;
    else if type='CLABPdfDoc' then
vbscripts='file2='||strip(type)||'.Open("||"&profiledir.&&Subj&subno"||' Lab.pdf)" ' ;
    ;
    VBOOrder=3 ;
    OUTPUT ;
  END ;

    vbscripts='Stack = eCRFPdfDoc.InsertPages(eCRFPdfDoc.GetNumPages-
1,CLABPdfDoc,0,CLABPdfdoc.GetNumPages,1)' ;
    VBOOrder=4 ;
    OUTPUT ;
    vbscripts='SaveStack = eCRFPdfDoc.Save(1,"||"&profiledir.&&Subj&subno"||'
Profile.pdf)'' ;
    VBOOrder=5 ;
    OUTPUT ;
  DROP type ;
RUN ;

PROC SORT data=getstmt out=getstmt ;
  BY VBOOrder ;
RUN ;

FILENAME vbsout "&profiledir.PDF_Combine.vbs" ;

DATA generate ;
  SET getstmt ;
  FILE vbsout ;
  PUT vbscripts ;
RUN ;

OPTIONS XSYNC NOXWAIT ;
X %unquote( %str(%) "&profiledir.PDF_Combine.vbs" %str(%) ) ;
***** Clean directory ***** ;
X %unquote( %str(%) del " &profiledir.&&subj&subno Lab.pdf" %str(%) ) ;

  %end ;

%MEND loopsubjs ;
%loopsubjs ;

ODS LISTING ;

```